



Fiche pratique – SOMMEPROD

La fonction **SOMMEPROD** permet d'additionner des produits.

C'est surtout parce qu'on peut la détourner de son usage initial qu'elle est intéressante. En effet, elle peut remplacer un NB.SI.ENS ou un SOMME.SI.ENS dès Excel 2003, dans les versions qui ne comprennent pas encore cette fonction ou pour inventer n'importe quel type de calcul conditionnel puisque qu'il n'y aura de limite que votre seule imagination.

Usage initial

	A	B
1	5	2
2	4	3
3	3	5
4		
5		

Faire la somme des lignes de produits

=SOMMEPROD(A1:A3;B1:B3)

Chaque argument comprend la plage à multiplier avec les autres et dont on veut additionner le produit :

(A1xB1)+ (A2xB2)+ (A3xB3)

Usage en NB.SI.ENS détourné

	A	B
1	Vitré	19
2	Caen	20
3	Vitré	16
4	Vitré	20
5		

Dans cet exemple, on recherche les valeurs correspondant à « Vitré » qui sont supérieures à 18.

=NB.SI.ENS(A1:A4;"Vitré";B1:B4;">18")

=SOMMEPROD((A1:A4="Vitré")*(B1:B4>18))

On n'utilise plus le « ; » de séparation d'arguments mais la multiplication de critères **A1:A4="Vitré"** et **B1:B4>18**.

Usage en SOMME.SI.ENS détourné

	A	B
1	Vitré	19
2	Caen	20
3	Vitré	16
4	Vitré	20
5		

Dans cet exemple, on additionne les valeurs pour la ville de Vitré.

=SOMME.SI.ENS(B1:B4;A1:A4;"Vitré")

=SOMMEPROD((A1:A4="Vitré")*(B1:B4))

On multiplie le critère **A1:A4="Vitré"** par la plage **B1:B4**.

Le fait que la seconde plage ne comporte pas d'égalité

indique à SOMMEPROD qu'il doit la traiter comme une donnée (et, donc, l'additionner aux autres puisque c'est le rôle initial de la fonction).

En fait, dans l'exemple du NB.SI.ENS détourné, on utilise aussi le même principe. On aurait pu écrire :

=SOMMEPROD((A1:A4="Vitré")*(B1:B4>18)*1)



Fiche pratique – SOMMEPROD

On peut donc définir n'importe quel type de calcul conditionnel en respectant cette écriture :

**=SOMMEPROD((Plage_Critère1=Critère1)* (Plage_Critère2=Critère2)...
* (Plage_à_Calculer1)opérateur(plage_à_Calculer2)....)**

	A	B	C	
1	Vitré	19	2	9,5
2	Caen	20	2	
3	Vitré	16	4	4
4	Vitré	20	1	20

Dans cet exemple, on invente un DIVISÉ.SI qui n'existe pas dans Excel :

=SOMMEPROD((A1:A4="Vitré") * ((B1:B4)/(C1:C4)))

On peut aussi imbriquer une fonction statistique comme MAX ou MIN dans le calcul :

**=SOMMEPROD(Fonction((Plage_Critère1=Critère1)* (Plage_Critère2=Critère2)...
* (Plage_à_Calculer1)opérateur(plage_à_Calculer2)....))**

	A	B
1	Vitré	19
2	Caen	20
3	Vitré	16
4	Vitré	20

Ici, par exemple, un MAX.SI.ENS (qui n'apparaît normalement que dans Excel 2016) :

=MAX.SI.ENS(B1:B4; A1:A4; "Vitré")

=SOMMEPROD(MAX((A1:A4="Vitré")*(B1:B4)))

Pour résumer l'articulation de cette fonction, écrivons-la par blocs :

Classique :

=SOMMEPROD(Bloc_valeurs1;Bloc_Valeurs2)

Chaque valeur est multipliée par celle des autres blocs dans sa ligne puis les produits sont cumulés.

Dénombrement :

=SOMMEPROD((Bloc_Critère1)*(Bloc_Critère2)...))

Chaque ligne cumulant l'ensemble des critères est comptée pour 1

Opération arithmétique :

=SOMMEPROD(((Bloc_Critère1)*(Bloc_Critère2)...) * ((Bloc_Valeurs1)opérateur(Bloc_Valeurs2)))

Chaque ligne correspondant aux critères verra son bloc de valeurs calculé selon l'opérateur choisi.

Fonction intégrée :

=SOMMEPROD(Fonction(((Bloc_Critère1)*(Bloc_Critère2)...) * (Bloc_Valeurs)))

Chaque ligne correspondant aux critères verra son bloc de valeurs calculé selon la fonction choisie.

L'ensemble des lignes sera additionné.

Attention à bien respecter le nombre et l'emplacement des parenthèses.